



**CENTRALE
NANTES**

INTRODUCTION AU CODE ÉLÉMENTS FINIS
CAST3M - CEA

Ecole Centrale de Pékin
Laurent GORNET



Centrale
Nantes

<http://www-cast3m.cea.fr>



DERNIÈRE MODIFICATION : MARS 2018 PÉKIN



**CENTRALE
NANTES**

- **Présentation de Cast3M**
- **Le langage Gibiane**
- **Travaux dirigés**
 - Résolution d'un système
 - Barre en traction
 - Vibration d'une poutre, vibration d'une plaque
 - Dynamique avec l'algorithme de Newmark
- **Compléments**
- **Description des objets Gibiane**

PAGE 2



■ Présentation de Cast3M

■ Le langage Gibiane

■ Travaux dirigés
 Résolution d'un système
 Barre en traction
 Vibration d'une poutre, vibration d'une plaque
 Dynamique avec l'algorithme de Newmark

■ Compléments

■ Description des objets Gibiane

PAGE 3



Logiciel de calcul par éléments finis en mécanique des structures et des fluides

- Résolution d'équations aux dérivées partielles par la méthode des éléments finis
- Basé sur un langage de commande : **Gibiane** (orienté objet)
- De nombreuses **fonctions élémentaires** (~ 1400)
- Procédure **PASAPAS** : solveur déterministe implicite (il en existe beaucoup d'autres ...)
- **Système complet** : solveur, pré-processeur, post-processeur, visualisation

PAGE 4

 **CENTRALE NANTES**

- **Mécanique des structures (historique)**
 Quasi-statique (non linéarités matériau, géométrie, conditions limites)
 Contact/frottement
 Flambage
 Dynamique (temporelle, modale, interaction fluide/structure)
 Rupture (XFEM, propagation dynamique, zones cohésives)
- **Thermique**
 Conduction, convection, rayonnement, changement de phase
- **Fluides**
- **Magnétostatique**
- **Diffusion multi espèces (loi de Fick)**
- **Couplage thermo-hygro-mécanique**



PAGE 5

 **CENTRALE NANTES**

- **Écrire un script Gibiane dans un fichier texte `toto.dgibi` et l'enregistrer dans un répertoire de travail**
- **Ouvrir une console / invite de commande**
 Sous **Linux** : nombreux émulateurs : `xterm`, `terminal`, `konsole`, ...
 Sous **Mac OS X** :
 Applications / Utilitaires / Terminal ou X11
 Sous **Windows** :
 Menu Démarrer / Tous les programmes / Accessoires / Invite de commande
- **Se placer dans le répertoire de travail et lancer Cast3M**
 En mode interactif : `castem17`
 En mode script : `castem17 toto.dgibi`

PAGE 6



■ Où télécharger Cast3M ?
<http://www-cast3m.cea.fr/index.php?page=dlcastem>

■ Pour quelles plateformes Cast3M est-il disponible ?
 → **Windows** (32 et 64 bits)
 → **Linux** (32 et 64 bits)
 → **Mac OS X** (64 bits)

■ Quelles versions de Cast3M sont disponibles ?
 Version Utilisateur : on ne peut pas modifier les sources
 Version Développeur : on peut modifier les sources ;-)

■ Y-a-t-il une interface graphique avec Cast3M ?
Non, Cast3M fonctionne uniquement en ligne de commande.
 Les instructions en langage Gibiane sont écrites dans un fichier texte d'extension .dgibi

■ Cast3M est-il gratuit ?
Oui, s'il est utilisé pour la recherche et l'enseignement
Non, s'il est utilisé de manière industrielle

PAGE 7



■ Le site web Cast3M : "the place to be"
<http://www-cast3m.cea.fr>

- *Présentation de Cast3M*
- *Formation*
- *Documentation (notices, manuels, sources, exemples)*
- *Fiches d'anomalie et de développement*
- *Téléchargements*
- *Contact : support Cast3M*
- *Communauté : liste de diffusion, club Cast3M*

PAGE 8

**CENTRALE
NANTES**

- **Langage destiné au calcul EF mais aussi un vrai langage de programmation**
 - Objets classiques (entiers, flottants, chaînes, logiques, tables ...)
 - Instructions conditionnelles
 - Boucles itératives
 - Sous structuration
 - Récurtivité ...
- **Langage interprété**
 - Le programme peut être exécuté dès que le script est modifié
 - Le programme peut être exécuté en mode interactif
- **Langage orienté objet**
 - Tout est traité comme un objet
 - Pas besoin de déclarer les variables ou de spécifier leur type
- **Mots clefs en français**
- **Programmation facile et rapide**



PAGE 9

**CENTRALE
NANTES**

- **Présentation de Cast3M**
- **Le langage Gibiane**
- **Travaux dirigés**
 - Résolution d'un système
 - Barre en traction
 - Vibration d'une poutre, vibration d'une plaque
 - Dynamique avec l'algorithme de Newmark
- **Compléments**
- **Description des objets Gibiane**

PAGE 10

 CENTRALE
NANTES

- **Ligne(s) de commande**
72 caractères max par ligne
504 caractères max par commande (soit 7 lignes pleines)
Se termine par un point virgule ;
Le symbole d'affectation est le signe égal =
- **Insensibilité à la casse** pour tous les noms d'objets
TOTO = 3.14 ;
A = 2. * tOTO ; ici la variable A vaut bien 6.28

sauf les chaînes de caractères 'blabla' ≠ 'BLABLA'
délimitées par des simples quotes mot1 = 'Salut a vous' ;
- **Fin du fichier de données**
par la commande **FIN ;** → arrêt de Cast3M
par une ligne vide ou un EOF → mode interactif
- **Ligne de commentaire : commence par ***
- **Lignes vides autorisées partout**

PAGE 11

 CENTRALE
NANTES

- **Pas de priorité des opérations (lecture de gauche à droite)**
 $1+2*3 = 9$ → penser à utiliser des parenthèses
 $1+(2*3) = 7$
- **Quelques interdictions**
Pas de **tabulations** → messages d'erreur incompréhensibles
Pas de **double quotes "**
- **Quelques recommandations**
Pas de **caractères spéciaux** (é, ç, ~, et autres œ)
Respecter une **indentation** (comme tout bon programmeur !)
Régler son **éditeur de texte**
coloration syntaxique, remplacement des tabulations
par des espaces, marquage de la colonne 72 ...
- **Quelques pièges classiques**
Point virgule à la 73^{ème} colonne → et la lecture de l'instruction
continue !
Mettre une **apostrophe** dans une chaîne de caractère
→ marque la fin de la chaîne

PAGE 12

 CENTRALE
NANTES

- **Définition**
Désigne toute **structure de données/résultats** munie d'un **type** (éventuellement d'un sous-type) et d'un **nom**
- **Nom des objets**
Donné par l'utilisateur
Limité à **8** caractères (a...z A...Z 0...9 _)
Pièges
Plus de 8 caractères : les surnuméraires sont ignorés
Utilisation du tiret – → interdit !
Caractères accentués é, è → interdit !
- **Type des objets**
Il existe plus de 40 types d'objets différents
Une liste des objets les plus utilisés est donnée à la fin de cette présentation ([lien](#))

PAGE 13

 CENTRALE
NANTES

- **Exemples (non exhaustif)**

```

OBJ1 = 3 ;                               type ENTIER
OBJ2 = 3.14 ;                             type FLOTTANT
OBJ3 = 'Comment ca va ?' ;               type MOT
OBJ4 = VRAI ;                             type LOGIQUE

poin1 = POIN 0. 0. ;                       type POINT
poin2 = POIN 1. 3. ;                       type POINT
OBJ5 = DROI 8 poin1 poin2 ;                 type MAILLAGE

LIST OBJ5 ;
MAILLAGE 3520406 : 8 élément(S) de type SEG2
0 sous-référence(s)
1ère ligne  numéro élément : 2ème couleur : 3ème... noeud(s)
           1      2      3      4      5      6      7      8
           DEFA  DEFA  DEFA  DEFA  DEFA  DEFA  DEFA  DEFA
           1      3      4      5      6      7      8      9
           3      4      5      6      7      8      9      2

```

PAGE 14



■ Définition
 Désigne tout **traitement** muni d'un **nom** (instruction Gibiane) qui construit un ou plusieurs **objets nouveaux** à partir d'un ou plusieurs objets existants

■ Noms des opérateurs
 Imposé à l'utilisateur
 Ce sont des instructions Gibiane
 Insensibles à la casse
 Cast3M ne lit que les 4 premiers caractères (**DROITE = DROI**)
 Quelques exceptions : forme abrégée
DROIT → **D** (ou **d**)
CERCLE → **C** (ou **c**)

PAGE 15



■ Exemples d'appel à un opérateur (invocation)

Cas courants (1 objet à gauche du =)
`obj1 = OPER obj2 ;`
`obj3 = OPER obj4 obj5 ;`
`obj6 = obj7 OPER obj8 obj9 ;`

Cas exceptionnels (plusieurs objets à gauche du =)
`obj1 obj2 obj3 = OPER obj4 obj5 ;`

Opérateur **"sans nom"** : création de POINTS
 en dimension 2 `Point1 = 0. 0. ;`
 en dimension 3 `Point1 = 0. 0. 0. ;`

PAGE 16

 **CENTRALE NANTES**

- **L'ordre des opérandes**
est **indifférent** si les opérandes sont de **type différents**
(sauf exception dans la documentation)

est **important** si plusieurs opérandes du **même type**
- **Surcharge d'un objet**
Toujours possible, l'ancien objet disparaît

```
A = 'Salut' ; → A est du type MOT
B = 28 ;
C = 3 ;
A = B**C ; → A est du type ENTIER et vaut 21952
```
- **Pièges**
Nom d'objet = nom d'opérateur → appel à l'opérateur impossible, sauf si on l'appelle en capitales entre quotes A = 'OPER' B C ;

Objet nommé c, C, d ou D !

PAGE 17

 **CENTRALE NANTES**

- **Définition**
Commande sans symbole d'affectation =
Ne crée pas de nouvel objet
- **Exemples**

```
OPTI 'DIME' 3 'ELEM' 'CUB8' ;
OPTI 'TITR' 'Maillage de la pièce' ;

DEPL mail1 'PLUS' (28. -0.3 20.03) ;
```

La directive **OPTI** est généralement la première instruction utilisée
Elle permet de fixer les **options générales** de Cast3M.
Exemples :
dimension de l'espace, éléments de maillage utilisés, taille de maille,
nom du fichier de sauvegarde, calcul axisymétrique, et bien d'autres ...

→ analogue à la fixation d'une variable d'environnement d'un système d'exploitation

PAGE 18



■ Définition
Ensemble nommé de commandes Gibiane muni d'une liste d'opérandes d'entrée et de sortie
Analogue à une subroutine Fortran ou à une fonction C

■ Nom des procédures
Comme un objet ordinaire (une procédure est un objet de type PROCEDUR)

■ Déclaration

```
DEBP ma_proc arg_e1*entier arg_e2*flottant ... arg_en*mchaml ;
    commande 1 ;
    commande 2 ;
    ...
    commande k ;
FINP arg_s1 arg_s2 ... arg_sm ;
```

PAGE 19



■ Invocation
Comme un opérateur ou une directive ordinaire

obj1 obj2 ... objm = **ma_proc** ent1 flot2 ... champn ;

■ Il existe des procédures pré-cablées dans Cast3M
Voir la liste : <http://www-cast3m.cea.fr/index.php?page=notices>

Les **@blabla**, préfixées par **@**, sont des procédures d'intérêt général

PASAPAS → calculs non linéaires
FLAMBAGE → calculs de flambage
DYNAMIC → calculs dynamiques
THERMIC → calculs thermiques
 ...

il y en a d'autres, à découvrir en naviguant dans la documentation

PAGE 20

 **CENTRALE
NANTES**

■ **Pièges**

FINP manquant
→ arrêt de Cast3M, message d'erreur parfois difficile à interpréter

FINP présent mais ; **manquant** ou au-delà position 72
→ arrêt de Cast3M, message d'erreur parfois difficile à interpréter

Invocation d'une procédure **avant** qu'elle ne soit **déclarée**
→ arrêt de Cast3M, message d'erreur de l'opérateur = parfois difficile à interpréter

PAGE 21

 **CENTRALE
NANTES**

■ **Débugage**

INFO OPER ;
→ affiche la notice d'un opérateur/directive/procédure

OPTI 'DONN' 5 ;
→ arrêt de la lecture du fichier .dgibi
→ lecture sur le terminal : **mode interactif**

OPTI 'DONN' 3 ;
→ reprise de la lecture du fichier .dgibi (là où il s'est arrêté)

LIST OBJ1 ;
→ liste le contenu de l'objet OBJ1

LIST 'RESU' OBJ1 ;
→ liste un **résumé** du contenu de l'objet OBJ1

OPTI 'DEBU' 1 ;
→ accès aux variables locales de la procédure

TRAC OBJ1 (OBJ2) ;
→ trace l'objet OBJ1 (maillage, champ, déformée, ...)

MESS 'Je passe ici !' ;
→ affiche un message

PAGE 22

 **CENTRALE
NANTES**

Création d'un modèle éléments finis

Organisation d'un calcul Elément-Finis (4 grandes étapes)

- **Choix de la géométrie et du maillage**
Définition des points lignes, surfaces, volumes, Discrétisation
- **Définition du modèle mathématique EF**
Définition du modèle EF (comportement matériau, types d'éléments)
Définition des propriétés matérielles (Young, masse volumique...)
Définition des propriétés géométriques (épaisseur des coques, moments quadratiques des poutres, ...)
Définition des conditions aux limites/chargements
Définition des conditions initiales
- **Résolution du problème discrétisé**
Calcul des matrices de rigidité et de masse pour chaque élément fini
Assemblage des matrices
Application des conditions limites/chargements
Résolution du système d'équations
- **Analyse et post-traitement des résultats**
Calcul de quantités locales (déplacement, contraintes, déformation, ...)
Calcul de quantités globales (énergie de déformation, ...)

PAGE 23

 **CENTRALE
NANTES**

- Présentation de Cast3M
- Le langage Gibiane
- **Travaux dirigés**
Résolution d'un système
Barre en traction
Vibration d'une poutre, vibration d'une plaque
Dynamique avec algorithme de Newmark
- Compléments
- Description des objets Gibiane

PAGE 24

CENTRALE NANTES Résoudre un système linéaire

* $2X + 11y + 12z = 0$
 * $X + y = 10$
 * $z = 0$

opti dime 2 ;
 pa = 1 0 ;
 mai = manu poi1 pa ;
 *

nom_inco = mots x y z ; nom_dual = mots sx sy sz ;

lig1 = prog 2 11 12 ;
 lig2 = prog 1 1 0 ;
 lig3 = prog 0 0 1 ;

rr = manu rigi mai nom_inco DUAL nom_dual 'QUEL' lig1 lig2 lig3 ;
 sec = manu chpo pa 3 'SX' 0. 'SY' 10. 'SZ' 0. ;
 sol = resou rr sec ; list sol ;

$$[K]\{q\} = \{f\}$$

$$\begin{bmatrix} 2 & 11 & 12 \\ 1 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{Bmatrix} x \\ y \\ z \end{Bmatrix} = \begin{Bmatrix} 0 \\ 10 \\ 0 \end{Bmatrix}$$

PAGE 25


CENTRALE NANTES

- Présentation de Cast3M
- Le langage Gibiane
- **Travaux dirigés**
 - Résolution d'un système
 - Barre en traction**
 - Vibration d'une poutre, vibration d'une plaque
 - Dynamique avec algorithme de Newmark
- Compléments
- Description des objets Gibiane

PAGE 26

CENTRALE NANTES

Principe des Travaux Virtuels pour une barre



$\forall \delta \bar{u} \quad \delta A = \delta W_{int} + \delta W_{ext}$

$$-\int_0^l \sigma_{xx} \delta \varepsilon_{xx} S dx \quad \text{avec} \quad \begin{cases} \delta \varepsilon_{xx} = \delta u_{,x} \\ \sigma_{xx} = E \varepsilon_{xx} = E u_{,x} \end{cases}$$

$$\forall \delta u \quad \int_0^l \rho S \ddot{u} \delta u dx = - \int_0^l E S u_{,x} \delta u_{,x} dx + \int_0^l f \delta u dx + F_0 \delta u_0 + F_l \delta u_l$$


$$-\delta E_d \quad 2E_d = \int_0^l E S (u_{,x})^2 dx$$

Formulation variationnelle

PAGE 27

CENTRALE NANTES

L'élément fini "Barre "




2 variables nodales \rightarrow approximation linéaire $u(x,t) = a_1 + a_2 x = \langle 1, x \rangle \begin{Bmatrix} a_1(t) \\ a_2(t) \end{Bmatrix}$

Identification nodale $\rightarrow u(x,t) = \langle 1 - \frac{x}{l_e}, \frac{x}{l_e} \rangle \begin{Bmatrix} u_i(t) \\ u_j(t) \end{Bmatrix}$ Maths

Approximation nodale $u = \langle N \rangle \{U_e\}$

$$N_1(x) = 1 - \frac{x}{l_e}$$

$$N_2(x) = \frac{x}{l_e}$$



Fonctions d'interpolation

PAGE 28

CENTRALE NANTES

L'élément fini "Barre "

Matrice raideur élémentaire associée à l'énergie de déformation élastique

$$2E_d = \int_0^{\ell_e} ES (u_{,x})^2 dx = \int_0^{\ell_e} u_{,x}^T ES u_{,x} dx$$

↓ $u_{,x} = \langle N_{,x} \rangle \begin{Bmatrix} u_i \\ u_j \end{Bmatrix}$

$$2E_d = \{U_e\}^T \int_0^{\ell_e} \langle N_{,x} \rangle^T ES \langle N_{,x} \rangle dx \{U_e\}$$

→ $[K_e] = \int_0^{\ell_e} \langle N_{,x} \rangle^T ES \langle N_{,x} \rangle dx$

$$[K_e] = \frac{ES}{\ell_e} \begin{bmatrix} 1 & -1 \\ -1 & 1 \end{bmatrix}$$

Matrice raideur élémentaire

CENTRALE NANTES

L'élément fini "Barre "

Travail des efforts extérieurs



$$\delta W = \int_0^{\ell} f \delta u dx + F_{ie} \delta u_i + F_{je} \delta u_j$$

↓ $\delta u = \langle N \rangle \{\delta U_e\}$

$$\delta W = \delta U_e^T \int_0^{\ell} N^T f dx + \delta U_e^T \begin{Bmatrix} F_{ie} \\ F_{je} \end{Bmatrix}$$

↓ $f = Cte$

$$\{\phi_e\} = \frac{f \ell_e}{2} \begin{Bmatrix} 1 \\ 1 \end{Bmatrix}$$

PAGE 30

CENTRALE NANTES

Dans le PTV en statique

$$\delta E_d = \delta W_{ext}$$

$$\forall e \quad \delta U_e^T [K_e] \{U_e\} = \delta U_e^T \{\phi_e + F_e\}$$

Assemblage $\int_D \dots = \sum_e \int_{D_e} \dots$
 $\sum_{noeud} \vec{F} = \vec{0}$

$$\forall \delta U \quad \delta U^T [K] \{U\} = \delta U^T \{\phi + F\}$$

Pour n valeurs

$$[K] \{U\} = \{\phi + F\}$$

Forces nodales $\left\{ \begin{array}{l} \text{Efforts donnés} \\ \text{Efforts de liaison (inconnues)} \end{array} \right.$

$$[K] \begin{Bmatrix} U_I \\ U_D \end{Bmatrix} = \begin{Bmatrix} F_D \\ F_I \end{Bmatrix} + \{\phi\}$$

Résolution par bloc

PAGE 31

CENTRALE NANTES

En post-traitement

Ayant $\{U_I\}$ On peut tracer la déformée de la structure
 $\{F_I\}$ Efforts aux appuis Ne doit pas être absurde
vérifient l'équilibre global de la structure

Efforts internes

Dans les éléments : on utilise les lois de comportement $N = ESu_{,x}$

$$N_e = \frac{ES}{\ell_e} (u_{je} - u_{ie})$$

Relation locale dans le repère de l'élément

Aux noeuds : on isole une partie de la structure \rightarrow équations d'équilibre

$$\forall e \quad [K_e] \{U_e\} = \{\phi_e\} + \begin{Bmatrix} F_{ie} \\ F_{je} \end{Bmatrix} \rightarrow \text{Efforts nodaux} \begin{Bmatrix} F_{ie} \\ F_{je} \end{Bmatrix}$$

CENTRALE NANTES

Application au treillis Plans

Changement de base en 2D

$$\begin{Bmatrix} \bar{u}_i \\ \bar{u}_j \end{Bmatrix} = \begin{bmatrix} C_\alpha & S_\alpha & 0 & 0 \\ 0 & 0 & C_\alpha & S_\alpha \end{bmatrix} \begin{Bmatrix} u_i \\ v_i \\ u_j \\ v_j \end{Bmatrix} \rightarrow \{\bar{U}_e\}^T = [P]\{U_e\}$$

Pour l'énergie

$$2E_d = \{U_e\}^T [P]^T \frac{ES}{\ell_e} \begin{bmatrix} 1 & -1 \\ -1 & 1 \end{bmatrix} [P]\{U_e\} = \{U_e\}^T [K_e]\{U_e\}$$

Pour les contraintes

$$N = \frac{ES}{\ell_e} (\bar{u}_j - \bar{u}_i) = \frac{ES}{\ell_e} \langle C_\alpha \quad S_\alpha \rangle \begin{Bmatrix} u_j - u_i \\ v_j - v_i \end{Bmatrix}$$

$$[K_e] = \frac{ES}{\ell_e} \begin{bmatrix} [A] & -[A] \\ -[A] & [A] \end{bmatrix}$$

avec $[A] = \begin{bmatrix} C_\alpha^2 & C_\alpha S_\alpha \\ C_\alpha S_\alpha & S_\alpha^2 \end{bmatrix}$

CENTRALE NANTES

Barre en traction

barreCEA2018.dgibi

Exemple simple : sur un élément barre (e)

```

OPTI 'DIME' 2 'ELEM' 'SEG2' ;
* Maillage
p1 = 0. 0. ;
p2 = 3. 0. ;
l1 = DROI 1 p1 p2 ;
* Modele et parametres
mo = MODE l1 'MECANIQUE' 'BARR' ;
ma = MATE mo 'YOUN' 210.E9 'NU' 0.3
      'SECT' 1.E-2 ;
* Matrice de rigidité principale
rig1 = RIGI mo ma ;
LIST rig1 ;
* Un petit blocage
b11 = BLOQ 'UX' p1 ;
LIST b11 ;
* Et un second membre associe
f1 = DEPI b11 1.2 ;
LIST f1 ;
    
```

matrices

$$\text{rig1} = \begin{bmatrix} k & 0 & -k & 0 \\ 0 & 0 & 0 & 0 \\ -k & 0 & k & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

inconnues primales

$$\begin{bmatrix} UX_1 \\ UY_1 \\ UX_2 \\ UY_2 \end{bmatrix}$$

inconnues duales

$$\begin{bmatrix} FX_1 \\ FY_1 \\ FX_2 \\ FY_2 \end{bmatrix}$$

$$b11 = \begin{bmatrix} 0 & 1 \\ 1 & 0 \end{bmatrix}$$

$$\begin{bmatrix} LX_3 \\ UX_1 \end{bmatrix}$$

$$\begin{bmatrix} FLX_3 \\ FX_1 \end{bmatrix}$$

$$\text{rig1 ET } b11 = \begin{bmatrix} k+0 & 0 & -k & 0 & 1 \\ 0 & 0 & 0 & 0 & 0 \\ -k & 0 & k & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 \end{bmatrix}$$

$$\begin{bmatrix} UX_1 \\ UY_1 \\ UX_2 \\ UY_2 \\ LX_3 \end{bmatrix}$$

$$\begin{bmatrix} FX_1 \\ FY_1 \\ FX_2 \\ FY_2 \\ FLX_3 \end{bmatrix}$$

$FLX_3 = 1,2$

PAGE 34

CENTRALE NANTES

Barre en traction

```

* Matrice raideur et résolution
dep1 = RESO (rig1 ET b11 ET b12) f1;
*****
* post-traitement
* déformée
def0 = DEFO dep1 L1 0. blanc;
EV1 = EVOL rouge CHPO dep1 Ux L1 ;
DESS ( EV1) TITRE 'déplacement U sur la hauteur d élément';
* Effort normal (N)
sig1 = SIGMA mo dep1 ma;
LIST Sig1; TITRE 'effort normal dans les éléments';
TRAC def0 sig1 mo;
* attention cet operateur fait la moyenne aux noeuds entre les éléments
CHPO1 = CHAN CHPO sig1 mo ;
EV2 = EVOL roug CHPO CHPO1 L1 ;
DESS EV2;
TITRE 'effort normal dans la barre';


```

PAGE 35


CENTRALE NANTES

Calcul statique des treillis de barres

Objectif : calculer les contraintes dans les barres, et la déformée des structures



Structure 1



Structure 2

- ✓ Quelle est la démarche à suivre avec le code EF Cast3M
- ✓ Calculer avec Cast3M les treillis avec un déplacement imposé.
- ✓ Calculer avec Cast3M les treillis avec une force imposée
 - OPERATEUR : FORCE .
 - Quelle force faut il imposer pour avoir le même résultat qu'en déplacement imposé

PAGE 36

CENTRALE NANTES

- Présentation de Cast3M
- Le langage Gibiane
- **Travaux dirigés**
 - Résolution d'un système
 - Barre en traction
 - Vibration d'une poutre**, vibration d'une plaque
 - Dynamique avec algorithme de Newmark
- Compléments
- Description des objets Gibiane

PAGE 37

CENTRALE NANTES

Principe des Travaux Virtuels pour les poutres

$$\forall \delta \vec{u} \quad \delta A = \delta W_{int} + \delta W_{ext}$$

$$\delta \vec{u} = \begin{cases} -y\delta\theta \\ \delta v \\ 0 \end{cases}$$

$$\forall \delta v \quad \int_0^\ell \rho S \ddot{v} \delta v \, dx = - \int_0^\ell EI v_{,xx} \delta v_{,xx} \, dx + \int_0^\ell f \delta v \, dx + F_o \delta v_o + F_\ell \delta v_\ell + M_o \delta \theta_o + M_\ell \delta \theta_\ell$$

$$- \int_0^\ell \sigma_{xx} \delta \varepsilon_{xx} \, S dx \quad \text{avec} \quad \begin{cases} \delta \varepsilon_{xx} = -y \delta v_{,x^2} \\ \sigma_{xx} = E \varepsilon_{xx} = -E y v_{,x^2} \end{cases}$$

$$- \delta E_d \quad 2E_d = \int_0^\ell EI (v_{,xx})^2 \, dx$$

Formulation variationnelle

CL du problème

PAGE 38

CENTRALE NANTES **L'élément fini « Poutre »**

4 variables nodales → approximation cubique

$$v(x,t) = \langle 1 \quad x \quad x^2 \quad x^3 \rangle \begin{Bmatrix} a_1(t) \\ a_2(t) \\ a_3(t) \\ a_4(t) \end{Bmatrix} \leftarrow \text{Maths}$$

Identification nodale

$$v(x,t) = \langle N_1 \quad N_2 \quad N_3 \quad N_4 \rangle \begin{Bmatrix} v_i(t) \\ \theta_i(t) \\ v_j(t) \\ \theta_j(t) \end{Bmatrix} \leftarrow \text{Physique}$$

Approximation nodale $v = \langle N \rangle \{U_e\}$

Fonctions d'interpolation

$$\begin{cases} N_1(s) = 1 - 3s^2 + 2s^3 \\ N_3(s) = 3s^2 - 2s^3 \end{cases} \quad \begin{cases} N_2(s) = \ell(s - 2s^2 + s^3) \\ N_4(s) = \ell(-s^2 + s^3) \end{cases} \quad s = \frac{x}{\ell}$$

PAGE 39

CENTRALE NANTES **Ecriture matricielle**

Approximation $v(x,t) = \langle w(x) \rangle \{q(t)\}$

Pondération $\delta v(x) = \langle w(x) \rangle \{\delta q\}$ «variation»

$$\forall \delta v \int_0^\ell \rho S \ddot{v} \delta v \, dx = - \int_0^\ell E I v_{,xx} \delta v_{,xx} \, dx + \int_0^\ell f \delta v \, dx + F_0 \delta v_0 + F_\ell \delta v_\ell + M_0 \delta \theta_0 + M_\ell \delta \theta_\ell$$

$$[M] = \int_0^\ell \langle w \rangle^T \rho \langle w \rangle \, dx \quad \{ \phi \} = \int_0^\ell \langle w \rangle^T f \, dx$$

$$[K] = \int_0^\ell \langle w_{,xx} \rangle^T E I \langle w_{,xx} \rangle \, dx \quad \{F\} = F_0 \langle w(0) \rangle^T + M_0 \langle w_{,x(0)} \rangle^T + F_\ell \langle w(\ell) \rangle^T + M_\ell \langle w_{,x(\ell)} \rangle^T$$

PTV $\Leftrightarrow \{ \delta q \}^T ([M] \{ \ddot{q} \} + [K] \{ q \}) = \{ \delta q \}^T \{ F + \phi \}$

Choix de $n \{ \delta q \}$

$$\Rightarrow [M] \{ \ddot{q} \} + [K] \{ q \} = \{ F + \phi \}$$

PAGE 40

CENTRALE NANTES Vibration d'une poutre et balayage en fréquence
poutre_vibration.dgibi

Options générales, paramètres, points, droites, maillage

■ **MAILLAGE**

```

OPTION ECHO 1 ;
affich = '0' ;
opti epsilon lineaire ;
GRAPH='N';
OPTION DIME 2 ELEM SEG2 ;
P1 = 0. 0. ;
P2 = 0. 10 ;
P3 = 0. 15.5;
L1 = (DROI P1 40 P2) ;
L2 = (DROI P2 20 P3) ;
L = L1 ET L2;
TRACE L ;

```



The diagram shows a vertical beam with points P1 at the bottom, P2 at the top, and P3 at the top. Segments L1 and L2 are defined between P1 and P2, and P2 and P3 respectively. The beam is divided into segments L1 and L2.

| PAGE 41

CENTRALE NANTES Vibration d'une poutre et balayage en fréquence

■ **PROPRIETES DE MATERIAUX**

```

MOD1 = MODEL L MECANIQUE ELASTIQUE POUT ;

d = 0.2;
SS = PI*d*d/4.;
IIZ = PI*d*d*d*d/64. ;

MODPOU = MODE L MECANIQUE ELASTIQUE POUT ;
MATPOU = MATE MODPOU RHO 7860. YOUN 2.1e11 NU 0.3 ;
CARPOU = CARA MODPOU 'SECT' SS 'INRZ' IIZ ;
MATPOU = MATPOU ET CARPOU ;
*

```

| PAGE 42

CENTRALE NANTES **Vibration d'une poutre et balayage en fréquence**

■ **CONDITIONS AUX LIMITES (UX UY RZ)**
 CLM1 = BLOQ 'DEPLA' P1 ;
 CLM2 = BLOQ 'ROTA' P1 ;
 CLM3 = BLOQ 'DEPLA' P2 ;
 CLM = CLM1 ET CLM2 ET CLM3 ;
 CLM = CLM1 ET CLM2 ;

■ **RIGIDITE ET MASSE**
 MASS1 = MASS MODPOU MATPOU ;
 RIG1 = RIGI MODPOU MATPOU ;
 RIG1 = RIG1 ET CLM ;

Ks = RIG1 ET CLM ;
 Majoutee = MASS 'DEPLA' 'ROTA' 4000 P3 ;
 Ms = MASS1 ET Majoutee ;

■ **ANALYSE MODALE DE LA POUTRE**
 Basr = VIBRATION 'INTERVALLE' 0.0001 100. Ks Ms 'TBAS' ;
 list basr;

Oscillations libres :

$$[M]\{\ddot{q}\} + [K]\{q\} = \{0\}$$
 Oscillateur à n degrés de liberté

$$\det[[K] - \omega^2 [M]] = 0$$

$$[[K] - \omega^2 [M]]\{x\} = 0$$

| PAGE 43

CENTRALE NANTES **Vibration d'une poutre et balayage en fréquence**

■ **RESULTATS ET DESSINS**
 Basm = Basr.'MODES' ; Nmod = DIME Basm-2 ;
 SI (EGA affich '0') ;
 MESSAGE ' POUTRE ' ;
 MESSAGE ' Mode Frequence ' ;
 MESSAGE '-----' ;
 REPETER LMod Nmod ;
 i=&LMod ;
 MESSAGE ' ' i ' ' Basm.i.'FREQUENCE' ;
 MOD = Basr.'MODES' . I .'DEFORMEE_MODAL' ;
 DEF0 = DEFO L1 MOD 0. ;
 DEF1 = DEFO L1 MOD 1 ROUGE ;
 TITRE ' ' i ' ' Basm.i.'FREQUENCE' ;
 TRAC (DEF0 ET DEF1) ;
 FIN LMod ;
 MESSAGE ' ' ;
 *DEF01 = ANIME 10 L1 MOD 5 ;
 *trac (defo1) OSCIL ;
 FINSI ;

| PAGE 44

CENTRALE NANTES **Vibration d'une poutre et balayage en fréquence**

Mode i $\{x\}_{\omega_i}$ Fréquence i $f_i = \frac{\omega_i}{2\pi}$

$\det[[K] - \omega^2 [M]] = 0$

$[[K] - \omega^2 [M]] \{x\} = 0$

Mode 1 Mode 2 Mode 3

$f_2 = 2.5378\text{Hz}$ $f_3 = 6.6526\text{Hz}$

| PAGE 45

CENTRALE NANTES **Vibration d'une poutre par balayage en fréquence**

$[M] \{ \ddot{q} \} + [K] \{ q \} = \{ F \} \cos(\omega t)$

BALAYAGE EN FREQUENCE


F = 1. ;
 Force1 = FORCE 'FX' F P3 ;
 ListeFrq = PROG 0. ;
 ufinal = PROG 0. ;
 *
 Frqi = 0. ; Frqf = 50. ; nfrq = 1000 ;
 dFrq = (Frqf - Frqi) / (nfrq - 1.) ;
 Frq = -1. * dFrq ;
 REPETER BALAYAGE nfrq ; \longrightarrow
 Frq = Frq + dFrq ;
 Gw2 = 4 * PI * PI * Frq * Frq ;
 Kvibra = RIG1 ET (-1. * Gw2 * MASS1);
 deplaF = RESO Kvibra Force1 ;
 ListeFrq = ListeFrq ET (PROG Frq) ;
 ufinal = ABS(ufinal/F) ET (abs((PROG (EXTRAIRE deplaF 'UX' P3))/F)) ;
 EVOLU_w = EVOL 'MANU' 'FREQUENCE' ListeFrq 'U' (ufinal/F) ;
 FIN BALAYAGE ;
 DESSIN EVOLU_w ;

$F_0 = F \cos(\omega t)$

$\{x\} = [[K] - \omega^2 [M]]^{-1} F$

Boucle sur la fréquence

| PAGE 46



CENTRALE NANTES

Calcul statique des treillis de poutres

Objectif : calculer les contraintes dans les poutres, et la déformée des structures

Structure 1

Structure 2

- ✓ Quelle est la démarche à suivre avec Cast3M
- ✓ Pour les poutres les liaisons entre les éléments sont des encastremets
- ✓ Calculer avec Cast3M les treillis avec un déplacement imposé en quasi-statique.
- ✓ Calculer avec Cast3M les treillis avec une force imposée en quasi-statique.
- ✓ Calculer les modes propres des structures

PAGE 48

 CENTRALE
NANTES

- Présentation de Cast3M
- Le langage Gibiane
- **Travaux dirigés**
 - Résolution d'un système
 - Barre en traction
 - Vibration d'une poutre, **vibration d'une plaque**
 - Dynamique avec algorithme de Newmark
- Compléments
- Description des objets Gibiane

PAGE 49

 CENTRALE
NANTES

Vibration d'une plaque

Objectif : créer un maillage d'une plaque et déterminer les pulsations et modes


1. placer des points maîtres (A B C D)
2. mailler le contour fermé,
3. puis la surface par remplissage
4. Calculer la rigidité et la masse
5. Appliquer les conditions limites
6. Calculer :fréquences et les modes

Théorie des Plaques minces
DKT

Les opérateurs Cast3M :

DROI
SURF
RIGI
MASS
BLOQ

VIBRATION



PAGE 50

CENTRALE NANTES

Vibrations libres d'une plaque

Mode 1

$$[M]\{\ddot{q}\} + [K]\{q\} = 0$$

Mode 2

$$\det[[K] - \omega^2 [M]] = 0$$

$$[[K] - \omega^2 [M]]\{x\} = 0$$

Orthogonalité des modes

$$\{x\}_s^T [K] \{x\}_r = \omega_r^2 \delta_{rs}$$

Raideur généralisée

$$\{x\}_s^T [M] \{x\}_r = \delta_{rs}$$

masse généralisée

Mode 3

| PAGE 51

CENTRALE NANTES

Vibration d'une plaque

■ Options générales et paramètres points , droites et surface

```

opti epsilon lineaire ;
opti echo 1; opti dime 3 elem TRI3;

a = 0. 0. 0.; b = 1. 0. 0.;
c = 0.2 0.6 0.; d = 0. 0.6 0.;
E = 0. 8. 0.;
N1 = 10; N2 = 20; N3 = 10;
*
AB = A DROI N1 B;
BC = B DROI N2 C;
CD = C DROI N3 D;
DA = D DROI N2 A;
sur1 = SURF(AB ET BC ET CD ET DA ) pla
'TRAC' sur1;
TRAC sur1;

```

| PAGE 52

 **CENTRALE
NANTES**

Vibration d'une plaque

■ **Modèle, paramètres, matrices de rigidité et de masse**

```

mod1 = MODE sur1 mecanique elastique DKT;
yop = 2.e11; nup = 0.28 ;
rho = 7860; ep = ((4.e-3)+(0.001/2.));
ep = ep * 1.83 ;

VT = ( mesu sur1 ) * ep ;
m = VT * rho;
MESS 'poid ' m ;

ma1 = MATE mod1 young yop nu nup rho rho;
ca1 = CARA mod1 epai ep;
*
rig1 = RIGI mod1 (ma1 et ca1);
mas1 = MASS mod1 (ma1 et ca1);
cl1 = BLOQUE DEPLA ROTA AB;
NMOD = 10;
RESUL = VIBR INTER 0.001 500. BASSE NMOD (RIG1 ET CL1) MAS1 TBAS;

```

| PAGE 53

 **CENTRALE
NANTES**

Vibration d'une plaque


■ **résultats**

```

REPETER BLOC1 NMOD;
fr1 = RESUL.MODES. &BLOC1 .FREQUENCE ;
num1 = RESUL.MODES. &BLOC1 .NUMERO_MODE ;
mgen1 = RESUL.MODES. &BLOC1 .MASSE_GENERALISEE ;
depi = RESUL.MODES. &BLOC1 .DEPLACEMENTS_GENERALISES;
VUX = depi. 1; VUY = depi. 2; VUZ = depi. 3;
Kgen1 = mgen1 * (2*pi*fr1)**2 ;
dep1 = RESUL.MODES. &BLOC1 .DEFORMEE_MODAL ;
TITRE ' MODE No ' num1 ' FREQUENCE ' fr1 ' Hz';
MESS 'MASSE GENERALISEE ' mgen1 ' RAIDEUR GENERALISEE ' Kgen1 ;
MESS 'DEPLACEMENTS GENERALISES X' VUX ' Y ' VUY ' Z ' VUZ;
def1 = defo dep1 sur1 ROUGE;
def0 = defo dep1 sur1 0. ;
trac (def1 et def0) ;
VZ = EXCO 'UZ' dep1 ;
trac VZ def1;
DEF01 = ANIME 15 sur1 dep1 VZ 1.;
trac DEF01 'OSCIL';
FIN BLOC1; resul;

```

| PAGE 54



CENTRALE NANTES

- Présentation de Cast3M
- Le langage Gibiane
- **Travaux dirigés**
 - Résolution d'un système
 - Barre en traction
 - Vibration d'une poutre
 - Dynamique avec algorithme de Newmark**
- Compléments
- Description des objets Gibiane

PAGE 56

CENTRALE NANTES Modélisation des systèmes continus

➤ exemple d'une barre en traction-compression

$u(x,t)$ déplacement d'une section droite

$N(x,t)$ effort normal

E module d'YOUNG

ρ masse volumique

✓ équations locales
 $\rho S \ddot{u} = N_{,x}(x,t)$

✓ lois de comportement
 $N(x,t) = E S u_{,x}(x,t)$

✓ conditions aux limites
 $u(0,t) = 0$
 $N(l,t) = F(t)$

✓ équations locales
 $\rho \ddot{u} - E u_{,xx}(x,t) = 0$

✓ conditions aux limites
 $u(0,t) = 0 \quad ES u_{,x}(l,t) = F(t)$

PAGE 57

CENTRALE NANTES Modélisation des systèmes continus

➤ exemple d'une barre en traction-compression : discrétisation du modèle continu

noeud i u_i l_e u_{i+1} noeud i+1

\xrightarrow{S}

$u(s,t) = \begin{bmatrix} 1 - \frac{s}{l_e} & \frac{s}{l_e} \end{bmatrix} \begin{bmatrix} u_i \\ u_{i+1} \end{bmatrix}$

assemblage

$Ec_e = \frac{1}{2} \begin{bmatrix} \dot{u}_i & \dot{u}_{i+1} \end{bmatrix} \rho S l_e \begin{bmatrix} 1/3 & 1/6 \\ 1/6 & 1/3 \end{bmatrix} \begin{bmatrix} \dot{u}_i \\ \dot{u}_{i+1} \end{bmatrix}$

$Ed_e = \frac{1}{2} \begin{bmatrix} u_i & u_{i+1} \end{bmatrix} \frac{ES}{l_e} \begin{bmatrix} 1 & -1 \\ -1 & 1 \end{bmatrix} \begin{bmatrix} u_i \\ u_{i+1} \end{bmatrix}$

➔

$Ec = \frac{1}{2} \{\dot{q}\}^T [M] \{\dot{q}\}$


$Ed = \frac{1}{2} \{q\}^T [K] \{q\}$

$\partial W = \{\partial q\}^T \{\Phi(t)\}$

Solution par superposition modale ou intégration en temps

$[M] \{\ddot{q}\} + [K] \{q\} = \{\Phi(t)\}$

PAGE 58



CENTRALE NANTES **Système linéaire différentiel 2^e ordre (1/3)**

```

opti dime 2 elem QUA4 ;
opti echo 0 ; opti epsilon lineaire ;
ldom = 50.;
hdom = 1.;
* Maillage
P1 = 0. 0. ; P2 = 0. hdom ;
P3 = ldom hdom; p4 = ldom 0. ;
nlelem = 20 *2 ; nhelem = 2*2;
L12 = DROI nhelem P1 P2 ;
L23 = DROI nlelem P2 P3 ;
L34 = DROI nhelem P3 P4 ;
L41 = DROI nlelem P4 P1 ;
dom = DALL L12 L23 L34 L41 ;
* trac dom;
* sortie resultats
P5 = (ldom /20.) hdom;
Pm = dom POINT proch p5 ;
Pm = P3 ;
dgauche = L12; ddroite = L34;
****
mod = dom MODE mecanique elastique isotrope
;
mat = mod MATE RHO 1. YOUNG 1. NU 0.3 ;

```

Code éléments finis Cast3M :
www-cast3m.cea.fr/

Modèle éléments finis 2D

PAGE 60

 **Système linéaire différentiel 2° ordre (2/3)**

```

* nombre de pas de temps
ndt = 200 ; dt = 1.;
dt = 0.8 * (ldom / nlelem); list dt;
* Newmark implicite
beta = 0.25 ; gamma = 0.5;
MESS 'gamma : ' gamma ' beta : ' beta ;
*
AA = BLOQ depla dgauche ;
F = table ;
F.0 = FORCE ddroite (0. 0.);
F.1 = FORCE ddroite (1. 0.);
KK = RIGI MOD MAT ;
MM = MASS MOD MAT ;
SS = MM ET ((beta * (dt ** 2.))* KK );
U = table ; dU = table ; ddU = table ;
S = table; TT = table ;
U.0 = MANU chpo dom 2 UX 0. UY 0. ;
dU.0 = MANU chpo dom 2 UX 0. UY 0. ;
ddU.0 = RESO ( SS ET AA ) ( F.0 ) ;
TT. 0 = 0 ;
*
S. 0 = SIGMA 'LINE' MOD MAT (U. 0);

```

Code éléments finis Cast3M :
www-cast3m.cea.fr/

PAGE 61

 **Système linéaire différentiel 2° ordre (3/3)**

```


n= 0 ;
Repete boucle ndt ;
F.(n+1) = F.1;
TT.(n+1) = (n+1) * dt;
* Prediction
dUpred = ( dU.n )+ ((dt * (1. -(gamma))) * ( ddU.n )) ;
Upred = ( U.n )+ (dt * ( dU.n ))
+ (( 0.5 * dt * dt * (1. - (2.* beta))) * (ddU.n));
ddU. (n+1) = RESOU (SS et AA ) ( ( F.(n+1)) - (KK * Upred)) ;
*Corrections
dU. (n+1) = dUpred +((dt * gamma)*(ddU. (n+1))) ;
U. (n+1) = Upred + (( dt * dt * beta) * (ddU. (n+1))) ;
S. (n+1) = SIGMA 'LINE' MOD MAT (U. (n+1));

*trac ( S.(n+1)) dom mod ;
MONTAGNE (EXCO UX ( ddU.(n+1))) dom (200. 200. ) 'CACHE' ;
* Contrainte
SXX = EXCO SMXX ( S.(n+1)) ;SXXP = CHAN 'CHPO' SXX MOD ;
SXXP = REDU SXXP L23 ;
MONTAGNE SXXP dom (200. 200. ) 'CACHE' ;
n= n+1 ;MESS ' n ' n ;
fin boucle ;

```

Algorithme de Newmark implicite
Code éléments finis Cast3M :

Algorithme officiel Newmark Cast3M
TAB1 = DYNAMIC TAB2 ;



Calcul dynamique des treillis de poutres

Objectif : calculer les contraintes dans les barres, et la déformée des structures

Structure 1: A cantilever beam of length l and height h fixed at the bottom left. A vertical force F is applied downwards at the top right corner. A coordinate system (\vec{x}_0, \vec{y}_0) is shown at the fixed end.

Structure 2: A truss structure with a horizontal base of length $a\sqrt{2}$. The left end is a pin support and the right end is a roller support. The top chord consists of two bars of length a meeting at a peak. A horizontal force F is applied to the right at the peak. A coordinate system (\vec{x}_0, \vec{y}_0) is shown at the left support.

Structure 2 is associated with a graph showing a step function, likely representing the applied force F over time.

- ✓ Quelle est la démarche à suivre avec le code EF Cast3M
- ✓ Faire le modèle élément finis de la barre en dynamique avec des barres
- ✓ Calculer avec Cast3M les treillis de poutre en dynamique avec **une force imposée F**
 OPERATEUR : FORCE , DYNAMIC

PAGE 64

 **CENTRALE NANTES**

- **Écrire dans un fichier texte**
→ en redirigeant l'impression vers un fichier

```

OPTI 'ECHO' 0 ;
OPTI 'IMPR' 42 'IMPR' 'mon_fichier.txt' ;

MESS 'J écris dans un fichier !' ;
MESS 4 8 15 16 23 42 ;
MESS ;

MESS 'N° iteration      Fibonacci' ;
FIBO1 = 0 ;
FIBO2 = 1 ;
MESS 0 '          ' FIBO1 ;
MESS 1 '          ' FIBO2 ;
REPE B1 15 ;
      FIBO2B = FIBO2 ;
      FIBO2 = FIBO1 + FIBO2 ;
      FIBO1 = FIBO2B ;
      MESS (&B1 + 1) '          ' FIBO2 ;
FIN B1 ;
OPTI 'IMPR' 6 ;

```

mon_fichier.txt

```

J écris dans un fichier !
4      8      15      16      23      42

N° iteration      Fibonacci
0                0
1                1
2                1
3                2
4                3
5                5
6                8
7                13
8                21
9                34
10               55
11               89
12              144
13              233
14              377
15              610
16              987

```

PAGE 65

 **CENTRALE NANTES**

Conclusion

- Utiliser le code EF Cast3M :
 - Choix de la géométrie et du maillage**
 - Définition du modèle mathématique EF**
 - Résolution du problème discrétisé**
 - Analyse et post-traitement des résultats**
- Programmer en Gibiane, le langage Cast3M
- Cast3M, un langage puissant associé à un code EF

Cast3M l'outil idéal pour résoudre les problèmes complexes

PAGE 66



FIN...

Questions ?

Code éléments finis Cast3M :
www-cast3m.cea.fr/



Gradient des déplacements	$\bar{\mathbf{H}} = \mathbf{Grad}(\bar{\mathbf{u}})$	$H_{ij} = \frac{\partial u_i}{\partial x_j} = u_{i,j}$
Déformations totales (Green Lagrange)	$\bar{\boldsymbol{\varepsilon}} = \frac{1}{2}(\bar{\mathbf{H}} + \bar{\mathbf{H}}^T + \bar{\mathbf{H}}^T \cdot \bar{\mathbf{H}})$	
Partition des déformations	$\bar{\boldsymbol{\varepsilon}} = \bar{\boldsymbol{\varepsilon}}^e + \bar{\boldsymbol{\varepsilon}}^{th}$	
Déformations thermiques	$\bar{\boldsymbol{\varepsilon}}^{th} = \alpha \Delta T \bar{\boldsymbol{\delta}}$	
Loi de comportement	$\bar{\boldsymbol{\sigma}} = \mathbf{C} : \bar{\boldsymbol{\varepsilon}}^e = \mathbf{C} : (\bar{\boldsymbol{\varepsilon}} - \bar{\boldsymbol{\varepsilon}}^{th})$	

avec :	$\bar{\mathbf{u}}$ déplacement $\bar{\boldsymbol{\varepsilon}}$ déformation totale $\bar{\boldsymbol{\varepsilon}}^e$ déformation élastique $\bar{\boldsymbol{\delta}}$ tenseur identité \mathbf{C} tenseur de rigidité (ordre 4) α dilatation thermique ΔT élévation de température	$\bar{\boldsymbol{\sigma}}$ contrainte $\bar{\boldsymbol{\varepsilon}}^{th}$ déformation thermique
--------	---	---

PAGE 68

CENTRALE NANTES

Hypothèse contraintes plane :

$$\bar{\sigma} = \begin{bmatrix} \sigma_{11} & \sigma_{12} & 0 \\ \sigma_{12} & \sigma_{22} & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad \bar{\varepsilon} = \begin{bmatrix} \varepsilon_{11} & \varepsilon_{12} & 0 \\ \varepsilon_{12} & \varepsilon_{22} & 0 \\ 0 & 0 & \varepsilon_{33} \end{bmatrix}$$

Notation vectorielle :

$$\{\sigma\} = \begin{Bmatrix} \sigma_{11} \\ \sigma_{22} \\ \sigma_{12} \end{Bmatrix} \quad \{\varepsilon\} = \begin{Bmatrix} \varepsilon_{11} \\ \varepsilon_{22} \\ 2\varepsilon_{12} \end{Bmatrix} \quad \text{avec : } \varepsilon_{33} = -\frac{\nu}{E}(\varepsilon_{11} + \varepsilon_{22})$$

3 composantes indépendantes

Loi de Hooke :

$$\begin{Bmatrix} \sigma_{11} \\ \sigma_{22} \\ \sigma_{12} \end{Bmatrix} = \frac{E}{1-\nu^2} \begin{bmatrix} 1 & \nu & 0 \\ \nu & 1 & 0 \\ 0 & 0 & \frac{1-\nu}{2} \end{bmatrix} \begin{Bmatrix} \varepsilon_{11} \\ \varepsilon_{22} \\ 2\varepsilon_{12} \end{Bmatrix}$$

$$\{\sigma\} = [C]\{\varepsilon\} \quad [C] \text{ matrice de rigidité}$$


PAGE 69

CENTRALE NANTES

Interpolation des inconnues primales :

$$\{u(x)\} = [N(x)]\{q\}$$

en tout point (x, y) de l'élément
 $\{q\}$ sont les déplacements des nœuds de l'élément



Cas d'un élément TRI3 (6 ddl) :

$$\begin{Bmatrix} u_x(x) \\ u_y(x) \end{Bmatrix} = \begin{bmatrix} N_A(x) & 0 & N_B(x) & 0 & N_C(x) & 0 \\ 0 & N_A(x) & 0 & N_B(x) & 0 & N_C(x) \end{bmatrix} \begin{Bmatrix} q_1^A \\ q_2^A \\ q_1^B \\ q_2^B \\ q_1^C \\ q_2^C \end{Bmatrix}$$

Déformations (linéarisées) :

$$\{\varepsilon(x)\} = \begin{bmatrix} N_{A,x_1} & 0 & N_{B,x_1} & 0 & N_{C,x_1} & 0 \\ 0 & N_{A,x_2} & 0 & N_{B,x_2} & 0 & N_{C,x_2} \\ N_{A,x_1} & N_{A,x_2} & N_{B,x_1} & N_{B,x_2} & N_{C,x_1} & N_{C,x_2} \end{bmatrix} \{q\} \quad \text{avec } N_{A,x_i} = \frac{\partial N_A(x)}{\partial x_i}$$

$$\{\varepsilon(x)\} = [B(x)]\{q\} \quad [B(x)] \text{ opérateur gradient discrétisé}$$

PAGE 70

Énergie de déformation élastique élémentaire :

$$\begin{aligned}
 e_{def} &= \frac{1}{2} \int_{V_e} \bar{\sigma} : \bar{\varepsilon} \, dV \\
 &= \frac{1}{2} \int_{V_e} \{\varepsilon\}^T \{\sigma\} \, dV \\
 &= \frac{1}{2} \int_{V_e} \{q\}^T [B]^T [C] [B] \{q\} \, dV \\
 &= \frac{1}{2} \{q\}^T \underbrace{\left[\int_{V_e} [B]^T [C] [B] \, dV \right]}_{[k_e]} \{q\}
 \end{aligned}$$

V_e "volume" de l'élément fini
 $[k_e]$ matrice de rigidité élémentaire
 (calculé par intégration numérique)

Assemblage :

$$E_{def} = \frac{1}{2} \{Q\}^T \sum_e [k_e] \{Q\} = \frac{1}{2} \{Q\}^T [K] \{Q\}$$

$\{Q\}$ déplacements des nœuds du maillage
 Les matrices $[k_e]$ sont étendues sur tous les
 ddl du maillage avec des termes nuls
 $[K]$ matrice de rigidité assemblée

PAGE 71

Principe des travaux virtuels :

\bar{f}_v force volumiques imposées
 \bar{f}_s force surfaciques imposées

$$\begin{aligned}
 \int_{V_e} \bar{\sigma} : \delta \bar{\varepsilon}^* \, dV &= \int_{V_e} \bar{f}_v \delta \bar{u}^* \, dV + \int_{S_e} \bar{f}_s \delta \bar{u}^* \, dS \\
 \int_{V_e} \{\delta \varepsilon^*\}^T \{\sigma\} \, dV &= \int_{V_e} \{\delta u^*\}^T \{f_v\} \, dV + \int_{S_e} \{\delta u^*\}^T \{f_s\} \, dS \\
 \{\delta q^*\}^T \int_{V_e} [B]^T \{\sigma\} \, dV &= \{\delta q^*\}^T \left(\int_{V_e} [N]^T \{f_v\} \, dV + \int_{S_e} [N]^T \{f_s\} \, dS \right) \\
 \underbrace{\int_{V_e} [B]^T [C] [B] \, dV}_{[k_e]} \{q\} &= \underbrace{\int_{V_e} [N]^T \{f_v\} \, dV + \int_{S_e} [N]^T \{f_s\} \, dS}_{\{F_e\}}
 \end{aligned}$$

$$[k_e] \{q\} = \{F_e\}$$

$\{F_e\}$ forces nodales équivalentes

PAGE 72

 CENTRALE NANTES

On ajoute un terme au second membre :

$$\int_{V_e} [B]^T \{\sigma\} dV = \{F_e\}$$

$$\int_{V_e} [B]^T [C] \{\varepsilon - \varepsilon^{th}\} dV = \{F_e\}$$

$$\int_{V_e} [B]^T [C] \{\varepsilon\} dV = \{F_e\} + \underbrace{\int_{V_e} [B]^T [C] \{\varepsilon^{th}\} dV}_{\{F_{th}\}}$$

$$[k_e] \{q\} = \{F_e\} + \{F_{th}\}$$

PAGE 73

 CENTRALE NANTES

- En mécanique on résout le problème contraint suivant :

$$\begin{cases} [K] \{Q\} = \{F\} \\ [b] \{Q\} = \{Q_0\} \end{cases}$$
 $[b]$ matrice de blocage
 $\{Q_0\}$ valeurs imposées
- La fonction à minimiser est :

$$f\{Q\} = \frac{1}{2} \{Q\}^T \cdot [K] \cdot \{Q\} - \{Q\}^T \cdot \{F\}$$
 avec les contraintes : $[b] \{Q\} - \{Q_0\} = \{0\}$
- Le théorème de Lagrange nous dit qu'il existe des réels $\{\lambda\}$ tels que :

$$[K] \{Q\} - \{F\} + [b]^T \{\lambda\} = \{0\}$$
- On se ramène alors au système suivant :

$$\begin{bmatrix} K & b^T \\ b & 0 \end{bmatrix} \begin{Bmatrix} Q \\ \lambda \end{Bmatrix} = \begin{Bmatrix} F \\ Q_0 \end{Bmatrix}$$

PAGE 74

 **CENTRALE
NANTES**

■ Objets généraux

ENTIER	Nombre entier
FLOTTANT	Nombre réel
MOT	Chaine de caractères
LOGIQUE	Booléen (VRAI ou FAUX)
LISTENTI	Liste d'entiers
LISTREEL	Liste de réels
LISTMOT	Liste de mots (limités à 4 caractères)
TABLE	Ensemble d'objets de type quelconque et caractérisés par un indice de type quelconque
EVOLUTIO	Représentation d'un graphe d'une fonction, suite de couples (x ; f(x))

PAGE 75

 **CENTRALE
NANTES**

■ Objets pour le maillage

POINT	Coordonnées d'un point de l'espace + densité
MAILLAGE	Domaine de l'espace discrétisé

■ Objets pour le calcul

CHPOINT	N'importe quel type de données aux nœuds d'un maillage (flottants, logiques, champs, ...)
MMODEL	Association d'un maillage, d'une formulation EF et d'un comportement matériau Définit les inconnues physiques primales / duales ex : <i>déplacements / forces</i> <i>température / flux</i>

PAGE 76

CENTRALE NANTES

- Objets pour le calcul (suite)
 - MCHAML** N'importe quel type de données à l'intérieur des éléments d'un maillage (flottants, logiques, champs, ...)
Variétés des points supports (points de Gauss, centre de gravité, nœuds, ...)
 - RIGIDITE** Matrice de raideur, de masse, de conductivité, ...
Matrice couplant les inconnues physiques
 - CHARGEME** Description spatio-temporelle d'un chargement
- Objets pour le post traitement
 - VECTEUR** Pour visualiser plusieurs composantes d'un champ
 - DEFORMEE** Pour visualiser un maillage déformé

PAGE 77

CENTRALE NANTES


Essai de traction

- **Modèle, paramètres, matrices de rigidité et de masse**

```

opti echo 1 ;
* TRI3 TRI6 QUA4 QUA8
opti dime 2 elem QUA4 'MODE' PLAN CONT ;
*opti dime 2 elem QUA4 'MODE' PLAN DEFO ;
* Construction du Maillage
p1= 0.0 0.0;
p2= 1.0 0.0;
p3= 1.0 1.0;
p4= 0.0 1.0;
*
N1 = 20;
P1P2= DROI N1 p1 p2 ;
P2P3= DROI N1 p2 p3 ;
P3P4= DROI N1 p3 p4 ;
P4P1= DROI N1 p4 p1 ;
*
SU1 = DALLER P1P2 P2P3 P3P4 P4P1;
trac su1;

```



PAGE 78

 **Essai de traction**

```

* 1 Définition du modèle
MODL1 = MODE SU1 'MECANIQUE' 'ELASTIQUE' 'ISOTROPE' ;
*
* 2 Affectation des propriétés de matériau et épaisseur de 2mm
*
xyoun = 2.1e+11 ; xnu = 0.3 ; xrho = 78000.;
* en contrainte plane préciser l'épaisseur 'DIM3'
MAT1 = MATE MODL1 'YOUN' xyoun 'NU' xnu 'RHO' xrho 'DIM3' 2.;
*
* 3 Blocages mécaniques
clim1 = BLOQ P1P2 'UY' ;
clim2 = BLOQ P1 'UX' ;
*

```

| PAGE 79

 **Essai de traction**

```

* 4 Définition du chargement
*
FOR = PRES MASS MODL1 -1. P3P4;
list for;
* passage en chargement tangentiel
* Pression vecteur (a,b): le tangent c'est (-b,a)
*F1 = EXCO 'FY' FOR 'FX'; F1 = -1.*F1;
*F2 = EXCO 'FX' FOR 'FY';
*FOR = F1 ET F2;
* vérification de la valeur du chargement
CHPO1 = RESU FOR ;
list CHP01;
*
* 5 Matrice de rigidité et de masse
RIG1 = RIGI MODL1 MAT1 ;

```

| PAGE 80


CENTRALE NANTES

Essai de traction

```

* Résolution lineaire de  $U = (K-1) F$ 
CLTOT = clim1 et clim2 ;
DEP1 = RESO (RIG1 et CLTOT) FOR ;
LIST DEP1;
*
* Calcul de la réaction
REA = REAC DEP1 CLTOT ;
* vérification de la valeur de la réaction
CHPO2 = RESU REA ; list CHPO2;
* visualisation du chargement et de la réaction
VEC3 = VECT REA 1 FX FY ROUGE ;
VEC2 = VECT FOR 1 FX FY JAUNE;
TRAC SU1 (vec2 et vec3) ;
* calcul et visualisation des contraintes
SIG1 = SIGM MODL1 MAT1 DEP1 ;
TRAC SIG1 SU1 MODL1 ;
*
DEF1 = DEFO SU1 DEP1 1000000000. ROUGE ;
DEF0 = DEFO SU1 DEP1 0. BLEU ; TRAC (DEF0 ET DEF1) ;

```



| PAGE 81

CENTRALE NANTES

Essai de traction

```

* extraire une composante
SM22 = EXCO SIG1 'SMYY';
TRAC SM22 DEF1 MODL1;
* dessiner une composante sur une ligne
SM22P = CHANGE 'CHPO' SIG1 MODL1 ;
*
EVOL1 = EVOL ROUGE 'CHPO' SM22P 'SMYY' P1P2 ;
EVOL2 = EVOL ROUGE 'CHPO' SM22P 'SMXX' P1P2 ;
DESS evol1 'MIMA'; DESS evol2 'MIMA';
*
*les directions principales
* 'INFE' 'MOYE' 'SUPE'
sip2 = PRIN SIG1 MODL1 MAT1 ;
COT = CONT SU1;
trac sip2 MODL1 COT ;
*
VEC1 = VECT sip2 MODL1 0.05;
* directions principales sans maillage
trac VEC1 COT ;
*

```

| PAGE 82



Essai de traction

```
* * calcul des déformations
*
EPS1 = EPSI  MODL1 DEP1 ;
TRAC EPS1 DEF1 MODL1;
*
* déplacement UY sur la ligne P3P4
EVOL1 = EVOL  BLEU  'CHPO' DEP1 'UY'  P3P4 ;
DESS  evol1 'MIMA';
*
```

| PAGE 83